
PROCEDURE QueryUser

LPARAMETERS lLogoff

LOCAL I, cPath, cFile, nCount, cTexte, cCommande
LOCAL cIdSession, cUsers, cCurrentUser

IF NOT EMPTY(GETENV("CLIENTNAME")) && C'est une session TSE

   ** On lance la commande QUERY USER
   M.cPath = ADDBS(SET("DIRECTORY"))
   STRTOFILE("QUERY USER >TSEUSERS.TMP", M.cPath+"TseUsers.bat")
   M.cFile = M.cPath + "TSEUSERS.TMP"   
   DELETE FILE (M.cFile)
   RUN /N2 TSEUSERS.BAT
   
   ** On attend que le fichier soit créé
   M.nCount = 0
   DO WHILE NOT FILE(M.cFile) AND M.nCount < 30
      =INKEY(0.3)
      M.nCount = M.nCount + 1
   ENDDO
   DELETE FILE (M.cPath+"TseUsers.bat")
   
   IF FILE(M.cFile)
      ** On attend la désignation de l'utilisateur en cours
      M.cTexte = ''
      M.nCount = 0
      DO WHILE AT('>', M.cTexte) = 0 AND M.nCount < 30 && Pas de current user
         =INKEY(0.3)
         M.cTexte = FILETOSTR(M.cFile)
         M.nCount = M.nCount + 1
      ENDDO
      DELETE FILE (M.cFile)
      IF NOT EMPTY(M.cTexte)
         ** On recherche le nom de l'utilisateur en cours
         STORE '' TO M.cUsers, M.cIdSession
         M.nCount = ALINES(aLignes, M.cTexte)
         FOR I = 2 TO M.nCount
             IF NOT EMPTY(aLignes(I))
                M.cUser = ALLTRIM(SUBSTR(aLignes(I),1,20))
                IF LEFT(M.cUser,1) = '>'
                   M.cCurrentUser = ALLTRIM(M.cUser)
                ELSE
                   M.cUsers = M.cUsers + IIF(EMPTY(cUsers),'>',",>") + M.cUser
                   M.cIdSession = M.cIdSession + ',' + SUBSTR(aLignes(I),44,3)
                ENDIF
             ENDIF
         ENDFOR

         ** L'utilisateur a ouvert plusieurs sessions sous le même nom
         IF OCCURS(M.cCurrentUser+',', M.cUsers+',') > 0
            Alerte("La session <" + UPPER(SUBSTR(M.cCurrentUser,2)) + ;
                   "> est déjà ouverte " + ;
                   STR(OCCURS(M.cCurrentUser+',', M.cUsers+','),1) + " fois")
            IF M.lLogoff
               IF NOT EMPTY(M.cIdSession) AND ;
                  Ask("On reprend la session précédente ?")
                  n = OCCURS(',', LEFT(M.cUsers, AT(M.cCurrentUser+',', M.cUsers+',')))+1
                  M.cCommande = "RUN /N2 TSCON " + ;
                                SUBSTR(M.cIdSession, AT(',',M.cIdSession,n)+1, 3)
                  &cCommande
                  =INKEY(5)
               ENDIF
               RUN /N2 LOGOFF
            ENDIF



            CANCEL
         ENDIF
      ENDIF
   ENDIF
ENDIF

*---------------------------------------------------------------- Alerte

PROCEDURE Alerte

LPARAMETERS cMessage

=MessageBox(IIF(VARTYPE(M.cMessage)='C',M.cMessage,TRANSFORM(M.cMessage)),48,"Attention")

RETURN

*---------------------------------------------------------------- Ask

FUNCTION Ask

LPARAMETERS cQuestion

RETURN (MessageBox(M.cQuestion,36,"Question")=6)


